Calcul de primitives

```
> load("integration.mc")$
```

```
/* integration.mc
   Quelques macros pour présenter des calculs d'intégrales.
   JM Sarlat - 2003 - http://melusine.eu.org/syracuse/maxima/
*/

/* Présentation d'une intégrale définie                         */
intdef(f,v,a,b) := 'integrate(f,v,a,b) = integrate(f,v,a,b);

/* Intégration en appliquant la relation de Chasles             */
intdefChasles(f,v,a,b,c) :=
        'integrate(f,v,a,c) = integrate(f,v,a,b) + integrate(f,v,b,c);

/* Intégration en utilisant un calcul de limite aux bornes      */
intconv(f,v,a,b) := 'integrate(f,v,a,b) = ldefint(f,v,a,b);

/* Intégration à l'aide d'une relation de Chasles avec calcul de limite
   aux bornes                                                   */
intconvChasles(f,v,a,b,c) :=
        'integrate(f,v,a,c) = ldefint(f,v,a,b) + ldefint(f,v,b,c);
```

```
/* Calcul de primitive                                              */
primitive(f,v) := 'integrate(f,v) = integrate(f,v);

/* Macro de simplification d'une expression trigo (1)               */
strig1(e,v) := ev(e,
        cos(v)+1=sin(v)/tan(v/2),
        tan(v/2)^2+1=2*tan(v/2)/sin(v),
        tan(v/2)^2-1=2*tan(v/2)/tan(v));

/* Macro de simplification d'une expression trigo (2)               */
strig2(e,v) := block([i],i:trigreduce(e),ev(i,
        2*cos(v)+2=tan(v/2)^2*(2-2*cos(v)),
        cos(2*v)=2*cos(v)^2-1));

/* Mise en oeuvre de simplications avec procédure finale            */
primitiveSimp(f,v,env,final) :=
        'integrate(f,v) = final(env(integrate(f,v),v));

/* Changement de variable                                           */
primitiveCV(f,v,eq,t) :=
        'integrate(f,v) = changevar('integrate(f,v),eq,t,v);
```

```
> intdef(1/(t*t^(1/3)),t,1,27);
```

$$\int_{1}^{27} \frac{1}{t^{\frac{4}{3}}}\,\mathrm{d}t = 2$$

```
> intdefChasles((1-abs(x-1))^3,x,0,1,2);
```

$$\int_{0}^{2} (1-|x-1|)^3\,\mathrm{d}x = \frac{1}{2}$$

```
> intdef(sqrt(2*x+1),x,0,4);
```

$$\int_{0}^{4} \sqrt{2\,x+1}\,\mathrm{d}x = \frac{26}{3}$$

```
> intdef(x^3*(1-x^2)^(5/2),x,0,1/2);
```

$$\int_{0}^{\frac{1}{2}} x^3\left(1-x^2\right)^{\frac{5}{2}}\,\mathrm{d}x = \frac{2}{63} - \frac{45\sqrt{3}}{3584}$$

```
> intdef(1/sqrt(9*x^2+3),x,0,1);
```

$$\int_{0}^{1} \frac{1}{\sqrt{9\,x^2+3}}\,\mathrm{d}x = \frac{\mathrm{Argsh}\left(\sqrt{3}\right)}{3}$$

```
> intdef(x^2*sin(x)*exp(x),x,0,%pi/2);
```

$$\int_{0}^{\frac{\pi}{2}} x^2\, e^x \sin x \, dx = \frac{\left(\pi^2-4\right)\, e^{\frac{\pi}{2}}}{8} + \frac{1}{2}$$

```
> primitiveSimp(log(sin(x))/cos(x)^2,x,strig1,expand);
```

$$\int \frac{\ln \sin x}{\cos^2 x} \, dx = -\tan x \ln \sin x - x$$

```
> intconv(1/(3*tan(x)+2),x,0,%pi/2);
```

$$\int_{0}^{\frac{\pi}{2}} \frac{1}{3 \tan x + 2} \, dx = \frac{3 \ln 3}{13} - \frac{3 \ln 2}{13} + \frac{\pi}{13}$$

```
> block(assume(a>0),intconvChasles(1/(a^2*cos(x)^2+sin(x)^2),x,0,%pi/2,%pi));
```

$$\int_{0}^{\pi} \frac{1}{\sin^2 x + a^2 \cos^2 x} \, dx = \frac{\pi}{a}$$

```
> primitiveSimp(cos(2*x)/(sin(x)+sin(3*x)),x,strig2,radcan);
```

$$\int \frac{\cos(2\,x)}{\sin(3\,x) + \sin x} \, dx = -\frac{\cos x \ln \tan\left(\frac{x}{2}\right) + 1}{4 \cos x}$$

```
> primitiveSimp((1-cos(2*x))/sin(3*x),x,strig2,radcan);
```

$$\int \frac{1-\cos(2\,x)}{\sin(3\,x)}\,dx = \frac{\ln(2\cos x+1)-\ln(2\cos x-1)}{2}$$

```
> primitive(1/(2*cosh(x)+sinh(x)+1),x);
```

$$\int \frac{1}{\operatorname{sh} x + 2\operatorname{ch} x + 1}\,dx = \frac{2\operatorname{Arctan}\left(\frac{6\,e^{x}+2}{2\sqrt{2}}\right)}{\sqrt{2}}$$

```
> f(x):=((x+1)^(1/2)-(x+1)^(1/3))/((x+1)^(1/2)+(x+1)^(1/3));
```

f(x):=((x+1)^(1/2)-(x+1)^(1/3))/((x+1)^(1/2)+(x+1)^(1/3));

```
> p:block(assume(t>0),primitiveCV(f(x),x,t-(1+x)^(1/6),t));
```

$$\int \frac{\sqrt{x+1}-(x+1)^{\frac{1}{3}}}{\sqrt{x+1}+(x+1)^{\frac{1}{3}}}\,dx = \int \frac{6\,t^{6}-6\,t^{5}}{t+1}\,dt$$

```
> rhs(p)=ev(rhs(p),nouns);
```

$$\int \frac{6\,t^{6}-6\,t^{5}}{t+1}\,dt = 12\ln(t+1) + \frac{5\,t^{6}-12\,t^{5}+15\,t^{4}-20\,t^{3}+30\,t^{2}-60\,t}{5}$$

```
> p:block(assume(t>0),primitiveCV(sqrt(x^3+1)/x,x,t^2-x^3-1,t));
```

$$\int \frac{\sqrt{x^3+1}}{x}\,\mathrm{d}x = 2\int \frac{t^2}{3\,t^2-3}\,\mathrm{d}t$$

```
> rhs(p)=ev(rhs(p),nouns);
```

$$2\int \frac{t^2}{3\,t^2-3}\,\mathrm{d}t = 2\left(-\frac{\ln(t+1)}{6}+\frac{\ln(t-1)}{6}+\frac{t}{3}\right)$$

```
> primitive((x+1)/sqrt(x*(1-2*x)),x);
```

$$\int \frac{x+1}{\sqrt{(1-2\,x)\,x}}\,\mathrm{d}x = \frac{5\,\mathrm{Arcsin}\,(4\,x-1)}{4\,\sqrt{2}} - \frac{\sqrt{x-2\,x^2}}{2}$$

```
> primitive(sin(2*x)*sinh(3*x),x);
```

$$\int \sin(2\,x)\,\mathrm{sh}\,(3\,x)\,\mathrm{d}x = \frac{e^{-3\,x}\left(\left(3\,e^{6\,x}+3\right)\sin(2\,x)+\left(2-2\,e^{6\,x}\right)\cos(2\,x)\right)}{26}$$

```
> p:primitiveCV(tan(x)^5,x,cos(x)-t,t);
```

SOLVE is using arc-trig functions to get a solution. Some solutions will be lost.

$$\int \tan^5 x\,\mathrm{d}x = -\int \frac{t^4-2\,t^2+1}{t^5}\,\mathrm{d}t$$

```
> rhs(p)=ev(rhs(p),nouns);
```

$$-\int \frac{t^4-2\,t^2+1}{t^5}\,\mathrm{d}t = -\ln t - \frac{4\,t^2-1}{4\,t^4}$$

```
> primitive(sin(x)/(2+tan(x)^2),x);
```

$$\int \frac{\sin x}{\tan^2 x+2}\,\mathrm{d}x = \mathrm{Arctan}\cos x - \cos x$$

```
> primitive(1/(cos(x)^4+sin(x)^4),x);
```

$$\int \frac{1}{\sin^4 x+\cos^4 x}\,\mathrm{d}x = \frac{\mathrm{Arctan}\left(\frac{2\tan x+\sqrt{2}}{\sqrt{2}}\right)}{\sqrt{2}} + \frac{\mathrm{Arctan}\left(\frac{2\tan x-\sqrt{2}}{\sqrt{2}}\right)}{\sqrt{2}}$$